Advanced Engineering Mathematics 1. Two-Hours Test May 14, 2019.

JE 13.5.19/ JKL 2019.05.14

PROBLEM 1

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=X->convert(Student[VectorCalculus][Del](X),Vector):
> with(LinearAlgebra):
```

A real function f of two real variables are given by

```
> f:=(x,y)->y/exp(x);
```

$$f := (x, y) \mapsto \frac{y}{e^x} \tag{1.1}$$

```
> f(x,y);
```

$$\frac{y}{e^x} \tag{1.2}$$

where $(x, y) \in \mathbb{R}^2$.

Question 1

The level curve corresponding to the level 0 is given by

$f(x, y) = \frac{y}{e^x} = 0 \Leftrightarrow y = 0$ and $x \in \mathbb{R}$, which is all of the x-axis.

The level curve corresponding to the level 1 is given by

$f(x, y) = \frac{y}{e^x} = 1 \Leftrightarrow \mathrm{y} = e^x$, $\mathrm{x} \in \mathbb{R}$, which is the graph for the exponential function.

```
> contourplot(f(x,y),x=-1..1,y=-1..2,contours=[0,1]);
```



Question 2

```
> diff(f(x,y),x);diff(f(x,y),y);
```

$$-\frac{y}{e^x}$$

$$\frac{1}{e^x} \quad \textbf{(1.2.1)}$$

$\nabla f(x, y) = (f'_x(x, y), f'_y(x, y)) = \left(-\frac{y}{e^x}, \frac{1}{e^x}\right)$.

$A = (0, 1)$, $B = (1, 2)$, $\mathbf{v} = \overrightarrow{AB} = (1, 1)$, $\mathbf{e} = \frac{\mathbf{v}}{|\mathbf{v}|} = \left(\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right)$, $\nabla f(A) = (-1, 1)$.

The directional derivative of f in the point A i n the direction towards the point B is then $f'(A; \mathbf{e}) = \mathbf{e} \cdot \nabla f(A) = 0$.

Since the gradient in the point is perpendicular to the level curve through the point, the result i not a surprise, since **e** is parallel to the tangent of the 1-level curve in A.

Let **e** be a unit vector drawn from the point A. The directional derivative of f in the point A in the direction of **e** is

$f'(A; \mathbf{e}) = \mathbf{e} \cdot \nabla f(A) = |\nabla f(A)| \cos(\varphi) = \sqrt{2}\cos(\varphi)$, where φ is the angle between **e** and ∇f (A).

For φ = 0 corresponding to **e** being in the direction of the gradient we get $\sqrt{2}$, that is the largest value of the directional derivative of f in the point A can attain.

A curve K in the (x, y)-plane is given by the parametric representation

```
> r:=u-><u,1+u-u^2>:
> r(u);
```

$$\begin{bmatrix} u \\ -u^2+u+1 \end{bmatrix} \quad \textbf{(1.3)}$$

where $u \in \mathbb{R}$.

Question 3

```
> ru:=diff(r(u),u);
```

$$ru := \begin{bmatrix} 1 \\ -2\,u+1 \end{bmatrix} \quad \textbf{(1.3.1)}$$

```
> gradf:=unapply(<diff(f(x,y),x),diff(f(x,y),y)>,[x,y]):
```

The gradient for f in the curve puoint $\mathbf{r}(u)$ is

```
> gradf(vop(r(u)));
```

$$\begin{bmatrix} -\dfrac{-u^2+u+1}{e^u} \\ \dfrac{1}{e^u} \end{bmatrix} \quad \textbf{(1.3.2)}$$

```
> g:=prik(gradf(vop(r(u))),ru);
```

$$g := -\frac{-u^2+u+1}{e^u}+\frac{-2\,u+1}{e^u} \quad \textbf{(1.3.3)}$$

```
> solve(g=0,u);
```

$$0,\ 3 \quad \textbf{(1.3.4)}$$

Thus the points on the curve K in which the gradient for f and the tangent vector $\mathbf{r}'(u)$ is orthogonal is the points $\mathbf{r}(0) = (0, 1)$ and $\mathbf{r}(3) = (3, -5)$.

The situation in the point (0, 1):

```
> P1:=plot(exp(x),x=-1..4,color=red):
> P2:=plot(1+x-x^2,x=-1..4,color=blue):
> P3:=plot(x+1,x=-1..1):
> display(P1,P2,P3,scaling=constrained,view=-5..3);
```



The 1-level curve and the curve K touches in the point A = (0, 1).

PROBLEM 2

```
> restart:with(plots):with(LinearAlgebra):with(student):
```

It is given that the symmetric matrix

```
> A:=<<2,-1>|<-1,2>>;
```

$$A := \begin{bmatrix} 2 & -1 \\ -1 & 2 \end{bmatrix} \quad \textbf{(2.1)}$$

has the eigenvalues 1 and 3 with corresponding eigenspaces E_1 = span{(1, 1)} and E_3 = span{(-1, 1)}.

Since **A** is symmetric, the two eigenspaces E_1 and E_3 are orthogonal.

A second-degree polynomial in two real variables are givetn by

```
> f:=(x,y)->2*x^2-2*x*y+2*y^2-4*x+2*y+2:
> f(x,y);
```

$$2x^2 - 2yx + 2y^2 - 4x + 2y + 2 \quad \textbf{(2.2)}$$

Question 1

In matrix form we have

$$f(x,y) = \begin{bmatrix} x & y \end{bmatrix} \begin{bmatrix} 2 & -1 \\ -1 & 2 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} + \begin{bmatrix} -4 & 2 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} + 2.$$

The matrix for teh included quadratic form

$$P(x,y) = \begin{bmatrix} x & y \end{bmatrix} \begin{bmatrix} 2 & -1 \\ -1 & 2 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix}$$

is exactly the given symmetric matrix **A**.
From the given facts it then follows that

```
> q1:=<1/sqrt(2),1/sqrt(2)>;
```

$$q1 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \end{bmatrix} \quad \mathbf{(2.1.1)}$$

is an orthonormal basis for E_1 and that

```
> q2:=<-1/sqrt(2),1/sqrt(2)>;
```

$$q2 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \end{bmatrix} \quad \mathbf{(2.1.2)}$$

where $\mathbf{q}2 = \widehat{\mathbf{q}1}$ is an orthonormal basis for E_3.
The eigenvectors ($\mathbf{q}1$, $\mathbf{q}2$) is then an orthonormal basis for $\mathbb{R}^2$ equiped with the ordinary scalar producst.
If we in the plane introduce an ordinary orthogonal coordinate system (O; **i**, **j**) then (O; **q**1, **q**2) is a new ordinary orthogonal coordinate system in the plane appearing from a rotation of the givnen coordinate system 45^{o} about O counter-clockwise.
If we put

```
> Q:=<q1|q2>;
```

$$Q := \begin{bmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{bmatrix} \quad \mathbf{(2.1.3)}$$

that is the cahnge of base matrix that in $\mathbb{R}^2$ changes from q-koordinates to standard e-coordinates and

```
> Lambda:=DiagonalMatrix([1,3]);
```

$$\Lambda := \begin{bmatrix} 1 & 0 \\ 0 & 3 \end{bmatrix} \quad \mathbf{(2.1.4)}$$

then **Q** is positive orthogonal and

$$\mathbf{Q}^{-1}\,\mathbf{A}\,\mathbf{Q} = \mathbf{Q}^{\mathrm{T}}\,\mathbf{A}\,\mathbf{Q} = \mathbf{\Lambda}$$

By application of the positive orthogonal substitution

$$\begin{bmatrix} x \\ y \end{bmatrix} = \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix},$$

that gives the relation between the (x, y)-coordinates and the new (x_1,y_1)-coordinates, the quadratic form in the new (x_1,y_1)-coordinates can be written int the reduced form

$$P(x, y) = P_1(x_1, y_1) = \begin{bmatrix} x_1 & y_1 \end{bmatrix} \mathbf{Q}^{\mathrm{T}} \mathbf{A}\, \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} = \begin{bmatrix} x_1 & y_1 \end{bmatrix} \begin{bmatrix} 1 & 0 \\ 0 & 3 \end{bmatrix} \begin{bmatrix} x_1 \\ y \end{bmatrix} = x_1^2 + 3y_1^2,$$

withpout product terms of the type $x_1 y_1$.

Question 2

```
> <-4|2>.Q.<x1,y1>+2;
```

$$-\sqrt{2}\, x1 + 3\sqrt{2}\, y1 + 2 \qquad \textbf{(2.2.1)}$$

```
> completesquare(x1^2+3*y1^2-sqrt(2)*x1+3*sqrt(2)*y1+2,[x1,y1])
  ;
```

$$3\left(y1 + \frac{\sqrt{2}}{2}\right)^2 + \left(x1 - \frac{\sqrt{2}}{2}\right)^2 \qquad \textbf{(2.2.2)}$$

By application of the positive orthogonal substitution from the previous question f can be written in the reduced form (cf the computation above)

$$f(x, y) = f_1(x_1,y_1) = x_1^2 + 3y_1^2 + \begin{bmatrix} -4 & 2 \end{bmatrix} \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} + 2 = x_1^2 + 3y_1^2 - \sqrt{2}\, x_1 + 3\sqrt{2}\, y_1 + 2 =$$

$$\left(x_1 - \frac{\sqrt{2}}{2}\right)^2 + 3\left(y_1 + \frac{\sqrt{2}}{2}\right)^2.$$

Thus $a = 1$, $b = 3$, $c = \frac{\sqrt{2}}{2}$ and $d = -\frac{\sqrt{2}}{2}$.

From

```
> Q.<sqrt(2)/2,-sqrt(2)/2>;
```

$$\begin{bmatrix} 1 \\ 0 \end{bmatrix} \qquad \textbf{(2.2.3)}$$

it follows that $(x_1, y_1) = (\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2})$ corresponds to $(x, y) = (1, 0)$.

$$f(1, 0) = f_1(\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2}) = 0.$$

Since

$$f(x, y) - f(1, 0) = f(x, y) = f_1(x_1,y_1) > 0$$

for all $(x_1,y_1) \neq (\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2})$ and thus for all $(x, y) \neq (1, 0)$ and only $= 0$

for $(x_1,y_1) = (\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2})$ and thus for $(x, y) = (1, 0)$,

then f has a proper local minimum 0 in $(1, 0)$. Actually 0 is the global minimum for f and the range is $f(\mathbb{R}^2) = [0; \infty[$.

PROBLEM 3

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector)
  :
  vop:=proc(X) op(convert(X,list)) end proc:
> with(LinearAlgebra):
```

Let a be a positive real number. A surface F in the (x, y, z)-space given by the parametric representation

```
> r:=(u,v)-><u^3*cos(v),u^3*sin(v),u>:
> r(u,v);
```

$$\begin{bmatrix} u^3 \cos(v) \\ u^3 \sin(v) \\ u \end{bmatrix} \qquad \textbf{(3.1)}$$

where $u \in [1/2; 1]$ and $v \in [0; a]$.

Question 1

Since the given parametric representation for the surface F has the form

$\mathbf{r}(u, v) = (g(u)\cos(v), g(u)\sin(v), h(u))$,

where $g(u) = u^3\cos(v) > 0$ for all $u \in [1/2; 1]$, $h(u) = u$ and $v \in [0; a]$,

then F according to page 11 in eNote 25 is a surface of revolution appearing from a rotation of the in the positive (x, z)-half plane situated profile curve with the parametric representation

$\mathbf{p}(u) = (u^3, 0, u)$, $u \in [1/2; 1]$,

through the angle v around the z-axis.

```
> plot([u^3,u,u=1/2..1],scaling=constrained,view=[0..1,0..1],
  title="Profilkurven i (x,z)-planen, der benyttes ved
  drejningen");
```



Question 2

```
> ru:=diff(r(u,v),u);
```

$$ru := \begin{bmatrix} 3\,u^2 \cos(v) \\ 3\,u^2 \sin(v) \\ 1 \end{bmatrix} \qquad \textbf{(3.2.1)}$$

```
> rv:=diff(r(u,v),v);
```

$$rv := \begin{bmatrix} -u^3 \sin(v) \\ u^3 \cos(v) \\ 0 \end{bmatrix} \qquad \textbf{(3.2.2)}$$

The normal vector of the surface is

```
> N:=simplify(kryds(ru,rv));
```

$$N := \begin{bmatrix} -u^3 \cos(v) \\ -u^3 \sin(v) \\ 3\,u^5 \end{bmatrix} \qquad \textbf{(3.2.3)}$$

The Jacobi function corresponding to **r** is

```
> Jacobi:=simplify(sqrt((prik(N,N)))assuming u>0,real;
```

$$Jacobi := u^3 \sqrt{9\,u^4+1} \quad \textbf{(3.2.4)}$$

Question 3

$f(x, y, z)$ is a mass density function in the (x, y, z)-space, where

$f(\mathbf{r}(u, v)) = \text{Jacobi}(u, v) = u^3\sqrt{1 + 9\,u^4}$ for all $u \in [1/2; 1]$.

The total mass on F is

$$\mathrm{M} = \int_F f\,\mathrm{d}\mu = \int_0^a \int_{\frac{1}{2}}^1 f(\mathbf{r}(u, v))\,\text{Jacobi}(u, v)\;\mathrm{d}u\,\mathrm{d}v = \int_0^a \int_{\frac{1}{2}}^1 \text{Jacobi}(u, v)\,\text{Jacobi}(u, v)\;\mathrm{d}u\,\mathrm{d}v$$

```
> integranden:=Jacobi*Jacobi;
```

$$integranden := u^6\left(9\,u^4+1\right) \quad \textbf{(3.3.1)}$$

```
> Int(Int(integranden,u=1/2..1),v=0..a)=int(int(integranden,u=
  1/2..1),v=0..a);
```

$$\int_0^a \int_{\frac{1}{2}}^1 u^6\left(9\,u^4+1\right)\mathrm{d}u\,\mathrm{d}v = \frac{151313\,a}{157696} \quad \textbf{(3.3.2)}$$

```
> a:=solve(151313*a*(1/157696)=1,a);
```

$$a := \frac{157696}{151313} \quad \textbf{(3.3.3)}$$

```
> evalf(a,5);
```

$$1.0422 \quad \textbf{(3.3.4)}$$

I.e. F attains the mass 1 for $\mathrm{a} = \frac{157696}{151313} \simeq 1{,}0422$.

PROBLEM 4

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector)
  :
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=X->convert(Student[VectorCalculus][Del](X),Vector):
  div:=V->VectorCalculus[Divergence](V):
  rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)
  end proc:
> with(LinearAlgebra):
```

A triangle T in the (x, y)-plane has the vertices $(0,0)$, $(1, 0)$ and $(1, 2)$.

```
> P1:=plot(2*x,x=0..1,filled=true):
> P2:=plot([1,y,y=0..2]):
> display(P1,P2,scaling=constrained,title="Trekanten T");
```



Question 1

A parametric representation for T is

```
> r(u,v):=<u,v*2*u>;
```

$$r(u, v) := \begin{bmatrix} u \\ 2\,v\,u \end{bmatrix} \tag{4.1.1}$$

where $u \in [0; 1]$ and v$\in[0; 1]$

Ω is the set of points in the (x, y, z)-space that lies vertically between T and the plane with the equation $z = 1$ (see the figure in the problem text).

Question 2

A parametric representaiton for Ω is

```
> r:=(u,v,w)-><u,v*2*u,w>:
> r(u,v,w);
```

(4.2.1)

$$\begin{bmatrix} u \\ 2\,v\,u \\ w \end{bmatrix} \quad \textbf{(4.2.1)}$$

where $u \in [0; 1]$, $v \in [0; 1]$ and $w \in [0; 1]$.

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>;
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 2\,v & 2\,u & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(4.2.2)}$$

```
> Jr:=Determinant(M);
```

$$Jr := 2\,u \quad \textbf{(4.2.3)}$$

that is ≥ 0, since $u \in [0; 1]$. The Jacobifunction corresponding to **r** is then

```
> Jacobi:=Jr;
```

$$Jacobi := 2\,u \quad \textbf{(4.2.4)}$$

A function $f := \mathbb{R}^3 \rightarrow \mathbb{R}$ is given by

```
> f:=(x,y,z)->x^2*y*z:
> f(x,y,z);
```

$$x^2\,y\,z \quad \textbf{(4.1)}$$

We onsider the vectorfield $\mathbf{V}(x, y, z) = \boldsymbol{\nabla} f(x, y, z)$.

```
> V:=(x,y,z)-><diff(f(x,y,z),x),diff(f(x,y,z),y),diff(f(x,y,z),z)
>:
> V(x,y,z);
```

$$\begin{bmatrix} 2\,x\,y\,z \\ x^2\,z \\ x^2\,y \end{bmatrix} \quad \textbf{(4.2)}$$

```
> divV:=div(V)(x,y,z);
```

$$divV := 2\,y\,z \quad \textbf{(4.3)}$$

```
> rot(V)(x,y,z);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \quad \textbf{(4.4)}$$

Question 3

$\partial\Omega$ is the closed surface of Ω with outward pointing unit normal vector. From Gauss' theorem we then get

$$\mathrm{Flux}(\mathbf{V}, \partial\Omega) = \int_{\Omega} \mathrm{Div}(\mathbf{V})\,\mathrm{d}\mu = \int_0^1 \int_0^1 \int_0^1 \mathrm{Div}(\mathbf{V})(\mathbf{r}(u, v, w))\,\mathrm{Jacobi}(u, v, w)\,\mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w$$

```
> integranden:=div(V)(vop(r(u,v,w)))*Jacobi;
```

$$integranden := 8\,v\,u^2\,w \quad \textbf{(4.3.1)}$$

```
> Int(Int(Int(integranden,u=0..1),v=0..1),w=0..1)=int(int(int
(integranden,u=0..1),v=0..1),w=0..1);
```

$$\int_0^1 \int_0^1 \int_0^1 8\,v\,u^2\,w\,\mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w = \frac{2}{3} \quad \textbf{(4.3.2)}$$

Question 4

A curve K, that runs along three edges of Ω from the origin to the point P = (1, 2, 1), is shown in red on the figur in the problem text.

According to Theorem 27.10, that also applies for a piecewise smooth curve, the wanted tangential curve integral is

$$\mathrm{Tan}(\boldsymbol{\nabla} f, K) = \int_K \boldsymbol{\nabla} f \bullet \mathbf{e}\, \mathrm{d}\mu = f(1, 2, 1) - f(0, 0\ 0) = 2 - 0 = 2.$$